
1 [bookmark: _Toc132734390][bookmark: _Toc122434485]Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2022-09_D23wk37 related to the title of this CR.
Contact:	Olivier Genoud
	olivier.genoud@etsi.org
2 [bookmark: _Toc122434488][bookmark: _Toc132734391][bookmark: _Toc295288959]Corrections required
2.1 [bookmark: _Toc106975634][bookmark: _Toc122623878][bookmark: _Toc109680267][bookmark: _Toc132734392][bookmark: _Toc325725665][bookmark: _Toc122434493][bookmark: _Toc295288970]Change 1
	Function name
	Function f_TC_13_5_1_EUTRA

	Reason for change
	A confirmation shall be requested in cases when there the MMI/AT command is followed by the starting of a timer

	Summary of change
	Requested a LOCAL_CNF when the function f_UT_RequestIMSCall is followed by a timer starting

	TTCN module
	AccessControl.ttcn



	  function f_TC_13_5_1_EUTRA() runs on EUTRA_PTC
  { // MTSI MO speech Call / SSAC / 0% access probability for MTSI MO speech call
    
    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;
    var NAS_MSG_Indication_Type v_NasInd;

    f_EUTRA_Init(c1, HANDLE_UM_DATA, USE_BIG_GRANTS);  /* @sic R5s170597 - Cleanup of UL grant Configuration: HANDLE_UM_DATA, USE_BIG_GRANTS sic@ */
    f_EUTRA_CellConfig_Def(eutra_Cell1);               /* @sic R5s170597 - Cleanup of UL grant Configuration: HANDLE_UM_DATA, USE_BIG_GRANTS moved to f_EUTRA_NAS_Init sic@ */
    
    f_EUTRA_CellInfo_SetSIB2_SSAC_BarringForMMTEL_Voice (eutra_Cell1,cs_AC_BarringConfig(p00, s64, '11111'B));
    f_EUTRA_SS_ConfigureSysinfo (eutra_Cell1);

    f_EUTRA_Preamble (eutra_Cell1, STATE2_IDLEUPDATE);
    f_EUTRA_TestBody_Set(true);
    
    // @siclog "Step 1" siclog@
    f_UT_RequestIMSCall(UT, px_IMS_CalleeUri, LOCAL_CNF_REQUIRED);
    
    // @siclog "Step 2" siclog@
    // Check: Does the UE transmit a RRCConnectionRequest message  within 10 s?.
    f_Delay(10.0);
    
    // @siclog "Step 3" siclog@
    f_EUTRA_CellInfo_SetSIB2_SSAC_BarringForMMTEL_Voice (eutra_Cell1, omit);
    f_EUTRA_ModifySysinfo (eutra_Cell1);
    
    // @siclog "Step 4" siclog@
    // Wait for 15 s (Note 1) to allow the new system information to take effect
    f_Delay(15.0);
    
    // @siclog "Step 5" siclog@
    f_UT_RequestIMSCall(UT, px_IMS_CalleeUri, LOCAL_CNF_REQUIRED);
    
    // @siclog "Step 6 - 7" siclog@
    // 34.229-1 TC 12.18: To verify that the UE does not send INVITE message within 10s because the timer Ty is running.
    f_Delay(59.0);
    
    // @siclog "Step 8 - 21" siclog@
    // Steps 1 to 14 of the generic test procedure for MTSI MO Speech call establishment in EUTRA: Normal Service (TS 36.508 4.5A.6.3-1).
    f_UT_RequestIMSCall(UT, px_IMS_CalleeUri); // @sic R5s0928 sic@
    f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell1, cr_EstablishmentCause_MoData);
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 5");
    
    // Steps 6 to 18 - MTSI MO call for EPS
    // @siclog "Step 6 - 18" siclog@
    v_NasInd := f_EUTRA_ServiceReq_WithoutRrcConnReq (eutra_Cell1, v_RRC_TI);
    
    f_EUTRA_Activate_SRB2_DRB_SendRrcMsg(eutra_Cell1,
                                         v_RRC_TI,
                                         v_NasInd.SecurityProtection.NasCount);
    
    f_EUTRA_Activate_SRB2_DRB_ReceiveRrcAndNasMsg(eutra_Cell1,
                                                  v_RRC_TI);
    
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    f_EUTRA_ActivateDedicatedEpsBearer_Speech(eutra_Cell1);
    
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]);
    
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    f_EUTRA_TestBody_Set(false);
    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED);
  }


2.2 Change 2
	Function name
	Function f_TC_13_5_1a_EUTRA

	Reason for change
	A confirmation shall be requested in cases when there the MMI/AT command is followed by the starting of a timer

	Summary of change
	Requested a LOCAL_CNF when the function f_UT_RequestIMSCall is followed by a timer starting

	TTCN module
	AccessControl_R12.ttcn




	  function f_TC_13_5_1a_EUTRA() runs on EUTRA_PTC
  { // MTSI MO speech Call / SSAC in Connected mode / 0% access probability for MTSI MO speech call
    
    f_EUTRA_Init(c1, HANDLE_UM_DATA, USE_BIG_GRANTS);  /* @sic R5s170597 - Cleanup of UL grant Configuration: HANDLE_UM_DATA, USE_BIG_GRANTS sic@ */
    f_EUTRA_CellInfo_SetSIB2_SSAC_BarringForMMTEL_Voice (eutra_Cell1,cs_AC_BarringConfig(p00, s64, '11111'B)); // @sic R5s160935 sic@
    f_EUTRA_CellConfig_Def(eutra_Cell1);               /* @sic R5s170597 - Cleanup of UL grant Configuration: HANDLE_UM_DATA, USE_BIG_GRANTS moved to f_EUTRA_NAS_Init sic@ */
    
    f_EUTRA_Preamble(eutra_Cell1, STATE2_IDLEUPDATE);
    f_EUTRA_RbEst_Def (eutra_Cell1); // @sic R5-185126 sic@
    f_EUTRA_TestBody_Set(true);
    
    // @siclog "Step 5" siclog@
    // Make the UE attempt a MTSI MO Speech call
    f_UT_RequestIMSCall(UT, px_IMS_CalleeUri, LOCAL_CNF_REQUIRED);
    
    // @siclog "Step 6" siclog@
    // Check: Does the UE transmit an INVITE message  within 10 s?
    f_Delay(10.0);

    // @siclog "Step 6AA" siclog@
    // @sic R5-162789 sic@
    f_EUTRA_CellInfo_SetSIB2_SSAC_BarringForMMTEL_Voice (eutra_Cell1, omit);
    f_EUTRA_ModifySysinfo (eutra_Cell1,-,RRC_CONNECTED); // @sic R5s160609 sic@

    // @siclog "Step 6AB" siclog@
    // Wait for 15 s (Note 1) to allow the new system information to take effect
    f_Delay(15.0); // @sic R5-162789 sic@
    
    // @siclog "Step 6A" siclog@
    // Make the UE attempt a MTSI MO Speech call
    f_UT_RequestIMSCall(UT, px_IMS_CalleeUri, LOCAL_CNF_REQUIRED); // @sic R5-160920 sic@
    // @siclog "Step 6B" siclog@
    // Wait for 10 s (Note 1) to allow the new system information to take effect.
    f_Delay(10.0); // @sic R5-160920 sic@
    
    // @siclog "Step 7" siclog@
    // Check: Does the UE transmit an INVITE message with in 10s?
    f_Delay(49.0); // @sic R5-160920 sic@ @sic R5-162789 sic@
    
    // @siclog "Step 8" siclog@
    f_UT_RequestIMSCall(UT, px_IMS_CalleeUri);
    
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]);
    
    // @siclog "Step 13 - 15" siclog@
    // Steps 12 to 14 of the generic procedure for MTSI MO speech call establishment (TS 36.508 table 4.5A.6.3-1).
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    f_EUTRA_ActivateDedicatedEpsBearer_Speech(eutra_Cell1);
    
    // @siclog "Step 23" siclog@
    f_EUTRA_508Check_ConnectedState (eutra_Cell1);
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]);
    
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    f_EUTRA_TestBody_Set(false);
    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED);
  }


2.3 Change 3
	Function name
	Function fl_EUTRA_9_2_1_2_1b_Body

	Reason for change
	A confirmation shall be requested in cases when there the MMI/AT command is followed by the starting of a timer

	Summary of change
	Requested a LOCAL_CNF when the function f_UT_RequestIMSCall is followed by a timer starting

	TTCN module
	NAS_Attach_UG.ttcn




	  function fl_EUTRA_9_2_1_2_1b_Body() runs on EUTRA_PTC
  {
...
      
      //@siclog "Step 11b2" siclog@
      //Check: Does the UE respond to paging on cell A with S-TMSI in GUTI-2 for PS domain?
      //(Generic Procedure TS36.508 6.4.2.4)
      f_EPS_508Check_NAS_Paging(eutra_CellA, ps);
      
      if (pc_VoLTE) { // @sic R5-122012 sic@
        //@siclog "Step 11b3a1" siclog@
        // An IMS call is initiated
        f_UT_RequestIMSCall (UT, -, LOCAL_CNF_REQUIRED);
        
        //@siclog "Step 11b3a2" siclog@
        // Check: Does the UE send RRC Connection Request message? (F, 4) @sic R5-120644 sic@
        v_TimerValue := f_EUTRA_SetTimerToleranceMin(eutra_CellA, nonProtocolTimer, 5.0);
        if (f_EUTRA_RRC_RRCConnectionRequest_Check(eutra_CellA, v_TimerValue)) {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 11b3a2");
        }
      }
      // send trigger to other RAT as test now finished
      f_EUTRA_SendTriggerToOtherRAT (); // @sic R5-112638 sic@
    }
  }


2.4 Change 4
	Function name
	Function fl_EUTRA_9_2_3_2_1b_Body

	Reason for change
	A confirmation shall be requested in cases when there the MMI/AT command is followed by the starting of a timer

	Summary of change
	Requested a LOCAL_CNF when the function f_UT_RequestIMSCall is followed by a timer starting

	TTCN module
	NAS_TrackingArea_UG.ttcn




	  function fl_TC_9_2_3_2_1b_Body() runs on EUTRA_PTC
  {
...
      
      //@siclog "Step 6b2" siclog@
      //Check: Does the UE respond to paging on cell B with S-TMSI in GUTI-2 for PS domain?
      //(Generic Procedure TS36.508 6.4.2.4)
      f_EPS_508Check_NAS_Paging(eutra_CellB, ps);
      
      if (pc_VoLTE) { // @sic R5-123651 sic@
        //@siclog "Step 6b3a1" siclog@
        // An IMS call is initiated
        f_UT_RequestIMSCall (UT, -, LOCAL_CNF_REQUIRED);
        
        //@siclog "Step 6b3a2" siclog@
        // Check: Does the UE send RRC Connection Request message? (F, 3)
        if (f_EUTRA_RRC_RRCConnectionRequest_Check(eutra_CellA, f_EUTRA_SetTimerToleranceMin( eutra_CellA, nonProtocolTimer, 5.0 ))) {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6b3a2");
        }
      }
      // Postamble
      f_EUTRA_TestBody_Set(false);
      f_EUTRA_Postamble(eutra_CellB, E1_IDLE, Combined); // @sic R5s110176 sic@
      // send trigger to UTRAN/GERAN side since Test case is now finished on E-UTRA side
      f_EUTRA_SendTriggerToOtherRAT(); // @sic R5-112638 sic@
    }
    
  }// end function fl_TC_9_2_3_2_1b_Body
  


2.5 Change 5
	Function name
	Function fl_TC_11_3_4_Body

	Reason for change
	A confirmation shall be requested in cases when there the MMI/AT command is followed by the starting of a timer

	Summary of change
	Requested a LOCAL_CNF when the function f_UT_InitiateECall is followed by a timer starting

	TTCN module
	IMS_ECall_UG.ttcn




	  function fl_TC_11_3_4_Body() runs on EUTRA_PTC
  {
    if (pc_CS_Em_Call_in_UTRA and px_RATComb_Tested == EUTRA_UTRA) {
      f_EUTRA_SendCoOrdToOtherRAT("UTRAN", cms_IRAT_Trigger);
      f_EUTRA_WaitForTriggerFromOtherRAT ("UTRAN");
    }
    else if(pc_CS_Em_Call_in_GERAN and px_RATComb_Tested == EUTRA_GERAN) {
      f_EUTRA_SendCoOrdToOtherRAT("GERAN", cms_IRAT_Trigger);
      f_EUTRA_WaitForTriggerFromOtherRAT("GERAN");
    }
    //@siclog "Step 6" siclog@
    f_UT_SwitchOnUE (UT, -, CNF_REQUIRED);

    //@siclog "Step 7" siclog@
    f_Delay (15.0);
    
    //@siclog "Step 8" siclog@
    f_UT_InitiateECall(UT, AUTOMATIC , -, LOCAL_CNF_REQUIRED);

    //@siclog "Step 9" siclog@
    if (f_EUTRA_RRC_RRCConnectionRequest_Check(eutra_CellA, 30.0, emergency)) { // @sic R5-230590 sic@
      f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Step 9");
    }
  }


2.6 Change 6
	Function name
	Function fl_TC_11_3_6_Body


	Reason for change
	A confirmation shall be requested in cases when there the MMI/AT command is followed by the starting of a timer

	Summary of change
	Requested a LOCAL_CNF when the function f_UT_InitiateECall is followed by a timer starting

	TTCN module
	IMS_ECall.ttcn




	  function fl_TC_11_3_6_Body() runs on EUTRA_PTC
  {
    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();
    var NAS_MSG_Indication_Type v_NasInd;
    var NAS_AttDetValue_Type v_DetachType;
    var GutiParameters_Type v_Guti1Params := f_EUTRA_CellInfo_GetGuti(eutra_CellA);
    var float v_EcallDeregisteredInactive := 60.0;
    var boolean v_Detached := false;
    timer t_EcallNoRRCConnReq := 120.0;
    timer t_Watchdog := 5.0;

    //@siclog "Step 1" siclog@
    // The UE is switched on.
    f_EUTRA_SwitchOnUEandStartIP(eutra_CellA, -, -, -, PDN_2, tsc_DRB2);// @sic R5s230115 sic@

    //@siclog "Step 2" siclog@
    // Wait for 60s for the UE to enter EMM-DEREGISTERED.eCALL-INACTIVE state
    f_Delay(v_EcallDeregisteredInactive);

    //@siclog "Step 3" siclog@
    // An eCall to URI for test service is initiated.
    f_UT_InitiateECall(UT, TEST , -, LOCAL_CNF_REQUIRED);

    //@siclog "Step 4" siclog@
    // Check: Does UE send an RRCConnectionRequest on Cell A within 120 seconds?
    t_EcallNoRRCConnReq.start;
    alt {
      [] t_EcallNoRRCConnReq.timeout {} // Continue the test
      [] SRB.check(receive(car_SRB0_RrcPdu_IND(?, cr_RRCConnectionRequest(cr_EstablishmentCause_Any)))) {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 2 - Unexpected RRCConnectionRequest");
        }
      }

    //@siclog "Step 5" siclog@
    // A manual eCall is initiated.
    f_UT_InitiateECall(UT, MANUAL);

    //@siclog "Step 6" siclog@
    // Check: Does the UE transmit an RRCConnectionRequest message with 'establishmentCause' set to 'emergency'?
    f_EUTRA_RRC_ConnectionRequest_Def(eutra_CellA, emergency);

    //@siclog "Step 7" siclog@
    // SS transmits an RRCConnectionSetup message.
    f_EUTRA_RRC_ConnectionSetup_Def (eutra_CellA, -, tsc_RRC_TI_Def);

    //@siclog "Step 8" siclog@
    // Check: Does the UE transmit an ATTACH REQUEST message to attach for emergency bearer services, together with a PDN CONNECTIVITY REQUEST message with 'Request type' set to 'emergency?
    v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(eutra_CellA,
                                                       cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),
                                                       cr_NAS_IndicationWithPiggybacking((tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected),
                                                                                          cr_508_ATTACH_REQUEST(tsc_EAttachType_Emergency,
                                                                                                                v_AdditionalUpdateType),
                                                                                          cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_Emergency,
                                                                                                                          f_GetPdnType())));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 8");

    //@siclog "Step 9-18" siclog@
    // Steps 5 to 14 of the generic test procedure for IMS Emergency call establishment in EUTRA: Limited Service (TS 36.508 subclause 4.5A.5.3-1).
    f_EUTRA_508_IMSEmergencyCallLimitedService_Steps7_16(eutra_CellA, v_NasInd); // @sic R5-227236 sic@

    //EXCEPTION: In parallel to the events described in steps 19 to 23 below, the behaviour in table 11.3.6.2-2 occurs.

    //@siclog "Step 19-23" siclog@
    // Steps 15 to 19 of the generic test procedure for IMS emergency call establishment in EUTRA: Limited Service (TS 36.508 subclause 4.5A.5.3-1).
    f_EUTRA_508_IMSEmergencyCallLimitedService_Steps17_19(eutra_CellA); // @sic R5-227236 sic@

    //@siclog "Step 24-27" siclog@
    // Release eCall over IMS using the generic procedure described in TS 34.229-1 [35] subclause C.33  @sic R5-196019 sic@
    //IMS call has been released.
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS2]);
    // The SS deactivates the EPS bearer. [This is the rest of C.33]
    f_EUTRA_DeactivateEPS_BearerContext (eutra_CellA,
                                         tsc_EpsDedicatedBearerId2); // there is only one PDN

    //EXCEPTION: Step 27Aa1 and 27Aa2 describes optional behaviour that depends on the UE implementation.
    v_DetachType :=  f_GetEDetachType(EPS_Only);
    t_Watchdog.start;
    alt {
      [] t_Watchdog.timeout {} // Continue the test
      [] //@siclog "Step 27Aa1" siclog@
         // The UE transmits a DETACH REQUEST message with the Detach type IE indicating "EPS detach" to regain normal service.
         SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA, //@sic R5-227236, @sic R5s230115 sic@
                                         cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                           cr_508_DETACH_REQUEST_MO(?,
                                                                                    f_GutiParameters2MobileIdentity(omit, v_Guti1Params),
                                                                                    cr_DetachType( '0'B, v_DetachType))))) {
          t_Watchdog.stop; // @sic R5s230115 sic@
          //@siclog "Step 27Aa2" siclog@
          // The SS responds the DETACH ACCEPT message.
          SRB.send (cas_SRB2_NasPdu_REQ (eutra_CellA, //@sic R5-227236, @sic R5s230115 sic@
                                         cs_TimingInfo_Now,
                                         cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                        cs_508_DETACH_ACCEPT)));
          v_Detached := true; // @sic R5s230115 sic@
        };
      }

    //@siclog "Step 29" siclog@
    // The SS releases the RRC connectionafter 5secs of Step 27..
    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);

    f_EUTRA_TestBody_Set(false); // @sic R5s230115 sic@

    if( v_Detached == false) {
      f_EUTRA_Postamble(eutra_CellA, E1_IDLE, EPS_Only);
    }
    else {
      f_EUTRA_Postamble(eutra_CellA, E4_DEREGISTERED);
    }
  }











